# Maintainer: Felix Yan <felixonmars@archlinux.org>

_hkgname=HsYAML-aeson
pkgname=haskell-hsyaml-aeson
pkgver=0.2.0.1
pkgrel=152
pkgdesc="JSON to YAML Adapter"
url="https://github.com/hvr/hsyaml-aeson"
license=('GPL')
arch=('loong64' 'x86_64')
depends=('ghc-libs' 'haskell-aeson' 'haskell-hsyaml' 'haskell-scientific'
         'haskell-unordered-containers' 'haskell-vector')
makedepends=('ghc' 'uusi')
source=(https://hackage.haskell.org/packages/archive/${_hkgname}/${pkgver}/${_hkgname}-${pkgver}.tar.gz)
sha512sums=('64602074e3e4424b7ff5055aa5aefac89e12127d56975c437189e434c168f129c6d5bb1ee6b0e22e201d0b25de574bb2a49f0fcd0ac1498250ae14920f89acf4')

prepare() {
    cd $_hkgname-$pkgver
    gen-setup
    uusi -u aeson -u vector $_hkgname.cabal
}

build() {
    cd $_hkgname-$pkgver    
    
    runhaskell Setup configure -O --enable-shared --enable-executable-dynamic --disable-library-vanilla \
        --prefix=/usr --docdir=/usr/share/doc/$pkgname --datasubdir=$pkgname --enable-tests \
        --dynlibdir=/usr/lib --libsubdir=\$compiler/site-local/\$pkgid --ghc-option=-fllvm
    runhaskell Setup build $MAKEFLAGS
    runhaskell Setup register --gen-script
    runhaskell Setup unregister --gen-script
    sed -i -r -e "s|ghc-pkg.*update[^ ]* |&'--force' |" register.sh
    sed -i -r -e "s|ghc-pkg.*unregister[^ ]* |&'--force' |" unregister.sh
}

check() {
    cd $_hkgname-$pkgver
    runhaskell Setup test --show-details=direct
}

package() {
    cd $_hkgname-$pkgver
    
    install -D -m744 register.sh "$pkgdir"/usr/share/haskell/register/$pkgname.sh
    install -D -m744 unregister.sh "$pkgdir"/usr/share/haskell/unregister/$pkgname.sh
    runhaskell Setup copy --destdir="$pkgdir"
    rm -f "${pkgdir}/usr/share/doc/${pkgname}/LICENSE.GPLv2"
}
